

## Aggregate Data and Memory

Compilers due: 2019-11-21 23:59:59 CET

In this lab you will update the compiler you built in earlier labs to the language **BX3**, which is an extension of **BX2** with support for aggregate data (arrays and structs) and the heap.

- This lab will be assessed.* It is worth 10% (i.e., 1 points) of your final grade. To get full credit, you must submit the final version of your compiler strictly before 2019-11-21 23:59:59 CET.

The complete grammar of BX3 can be found in `bx3_grammar.pdf`. The BX3 language is a strict superset of the BX2 language. It adds no new control structures, but does modify the type system and adds support for reading and writing from memory and structured data.

The BX2 type system consisted of exactly two types, `int64` and `bool`. BX3 extends that with two more type forms: pointers and arrays. Given a type  $\tau$ , the type  $\tau^*$  is the type of pointers to a chunks of memory that contain values of type  $\tau$ .

**ALLOCATION** To allocate a new block of heap memory holding a finite array of values of type  $\tau$ , BX3 adds a new `alloc` expression which take the type of the contents,  $\tau$ , and the number of elements in the array as parameters.

The expression `alloc  $\tau$ [ $n$ ]` allocates  $n$  (which must be  $> 0$  at runtime) contiguous blocks of size suitable to store values of type  $\tau$ ; this expression then itself has type  $\tau^*$ . The count argument  $n$  is mandatory: if you want to allocate a single slot, use 1 for  $n$ . The memory that is returned by `alloc` is 0'd out. Thus, the initial value stored in the allocated memory is whatever is represented with a sequence of 0-bytes.

[2019-11-15 14:50]

Given a pointer-valued expression  $e$  of type  $\tau^*$ , the expression  $*e$  *dereferences* the pointer to return the value of type  $\tau$  that is stored at the address  $e$ . Likewise,  $e[n]$  gets the element that is at slot  $n$  in the array allocated there. You may think of  $*e$  as a synonym for  $e[0]$ .

**ASSIGNMENT** The expression forms  $*e$  and  $e[n]$  behave like variables and can be assigned to like any other variable. In fact there are a collection of such expressions that we call *assignable*.<sup>1</sup>

$\langle \text{assignable} \rangle ::= \langle \text{variable} \rangle \mid "*" \langle \text{expr} \rangle \mid \langle \text{expr} \rangle "[" \langle \text{expr} \rangle "]"$

Assignment statements now allow setting any assignable to a value of a compatible type.

$\langle \text{assign} \rangle ::= \langle \text{assignable} \rangle "=" \langle \text{expr} \rangle ";"$

Note that all *assignable*s are *expr*essions.

$\langle \text{expr} \rangle ::= \dots \mid \langle \text{assignable} \rangle$

**REFERENCING** Any assignable of type  $\tau$  can be turned into a pointer (of type  $\tau^*$ ) by means of the unary referencing operator,  $\&$ .

$\langle \text{expr} \rangle ::= \dots \mid "&" \langle \text{assignable} \rangle$

This pointer can be used like any other pointer and can even be passed to other procedures and functions. However, note that pointers that point to the current stack frame will be automatically invalidated with the current call *returns*, so if these pointers are stored in data structures that persist past the lifetime of the current call then it is undefined what happens when those pointers are subsequently used. BX3 is not memory-safe: dereferencing invalid pointers can blow your stack and crash your program.

**NULL POINTERS** All pointers are allowed to be *null*, which is a special value that denotes an invalid pointer. It is a new kind of expression that can have any pointer type, i.e., the value *null* can be used to assign to any assignable of type  $\tau^*$ . It is a dynamic error to dereference *null*, which will immediately abort the program. From a representational standpoint, the bytes used to represent *null* are all 0-bytes.

$\langle \text{expr} \rangle ::= \dots \mid "null"$

Some examples:

```
proc f (p : int64*) {
  //
}

proc main() {
  var x = null : int64*; // can initialize int64* with null
  f(null);               // can use null in place of int64* args
  print null;            // outputs "null"
}
```

<sup>1</sup>In C/C++ these are known as *lvalues*.

**POINTER OPERATIONS** The only allowed operators on pointers are `==` and `!=`. We have the following table for the expression `p == q` where both `p` and `q` are of the pointer type  $\tau^*$ :

<code>p is null?</code>	<code>q is null?</code>	<code>p == q</code>
yes	yes	<b>true</b>
no	yes	<b>false</b>
yes	no	<b>false</b>
no	no	<b>true</b> iff <code>p</code> and <code>q</code> point to the same thing

## 2.2 Fixed-Size Arrays

For a literal non-negative number  $n \in \mathbb{N}$ , the type  $\tau[n]$  is the type of arrays of fixed size  $n$  containing elements of type  $\tau$ . The benefit of fixed-size arrays is that their size is known at compile time, and hence when they are themselves used in aggregate structures it is not necessary to use extra pointers.

$\langle \text{type} \rangle ::= \dots \mid \langle \text{type} \rangle \text{"["} \langle \text{number} \rangle \text{"}"}$

To access an element of an array, you would use square brackets for subscripting, just like for ordinary pointers. To allocate such an array on the heap, you would use an `alloc` expression as usual.

```
// allocate a single int64[10].
var x = alloc int64[10][1] : int64[10]*;

// allocate 20 blocks of int64[10]s.
var y = alloc int64[10][20] : int64[10]*;
```

Subscripting for fixed-size arrays is the same as for variable-sized arrays.

**ASSIGNMENT** In the assignment expression `l = e`; where `e` has type  $\tau[n]$ , the effect is to copy each of the  $n$  elements of `e` into `l`.

## 2.3 Structured Data (Optional)

BX3 also allows for structured data with named fields. Such types are introduced with the `struct` keyword and define a list of unique mappings from field names to the types of the fields.

$\langle \text{type} \rangle ::= \dots \mid \text{"struct"} \text{"{"} ( \langle \text{struct-field} \rangle ( \text{","} \langle \text{struct-field} \rangle^* \text{","} )? \text{"?"} ) \text{"}"}$   
 $\langle \text{struct-field} \rangle ::= \langle \text{variable} \rangle \text{":"} \langle \text{type} \rangle$

The size of a `struct` type is the sum of the sizes of the types of the various fields of the `struct`. Each field also has a fixed *offset* from the start of the header, determined by the sizes of all the field types that come before the given field in the `struct`.

To illustrate, consider the following `struct`:

```
struct { f1 : int64, f2 : int64*, f3 : int64[10], f4 : struct { g1 : bool } }
```

Its fields have the following sizes and offsets:

field	size (bytes)	offset (bytes)
f1	8	0
f2	8	8
f3	80	16
f4	8	96

The `struct` itself is 104 bytes in size.

**INDEXING AND ASSIGNMENT** To access a given field of a `struct` we use `.`-notation like in C. Given an expression `e` of type

```
struct { f1 :  $\tau_1$ , f2 :  $\tau_2$ , ..., fk :  $\tau_k$  }
```

the expression `e.fi` is an expression of type  $\tau_i$ . This notation is also assignable. As a standard simplification of the syntax for subscripting structs, we use the notation `e->f` to stand for `(*e).f`.

`<assignable> ::= ... | <expr> "." <variable> | <expr> "->" <variable>`

Like with arrays, assigning a variable of struct type copies all the fields of the struct.

## 2.4 Type Abbreviations and Recursive Types (Optional)

Note that the `struct` types of BX3 are anonymous. This means that the only way to create recursive types is to introduce *type names* to the language. We do this by means of type *abbreviations*, which are a new kind of top-level declaration.

`<program> ::= (<globalvar> | <function> | <procedure> | <type-abbrev>)*`  
`<type-abbrev> ::= "type" <variable> "=" <type> ";"`

Such an abbreviation introduces a new synonym for a given type.

Like other top-level declarations, type abbreviations are also mutually recursive with all the other type abbreviations; indeed, this is the only way we can define a recursive type such as a linked list, as the example below shows.

```
type list = struct {
  data : int64,
  next : list*,
};
```

To compute the size of a type in the presence of type abbreviations, we sometimes need to look up the definition of the abbreviation. In the occurrence of `list*` we did not need to look up `list` since pointer types always have size 8 bytes. However, if we were to write the following recursive type, then we would have a cyclicity in the size calculation.

```
type bad_list = struct {
  data : int64,
  next : bad_list,
};
```

Here, to compute the size of `bad_list` we will eventually need to know the size of `bad_list` itself. All types in BX3 are required to have sizes that can be determined without such cycles. The type `bad_list` above will therefore be considered to be ill-formed.

## 2.5 Further BX3 Language Notes

**SENDING AND RECEIVING AGGREGATE TYPES** Aggregate types (arrays and structs) may be used as desired in the body of a callable. However, it is *not allowed* to define a callable that has a parameter of aggregate type, nor is it allowed to return an aggregate type from a function. Note that you may send and receive *pointers* to aggregate types.

**YE OLDE FIBONACCI EXAMPLE** A CSE302 lab won't be complete without an updated Fibonacci example. One is provided in figure 3 that makes use of some of the new features of BX3. Note that it is far from comprehensive as tests go.

## 3 COMPILING BX3

### 3.1 Stack vs. Heap Allocation

A temporary or local variable of aggregate type (array or struct) can be allocated on the stack just like a variable of non-aggregate type. These variables do not have initializing expressions,<sup>2</sup> but are instead 0'd out. To reinforce this, we will use the syntax `= 0` for initializing expressions for aggregate types, even though the literal number 0 is usually of integral type. Examples:

```
// an array of 10 integers, all 0
var x = 0 : int64[10];

// a record with field f as 0, g as false (represented as 0)
var y = 0 : struct { f : int64, g : bool };
```

For heap-allocation, on the other hand, the new memory is guaranteed to be 0'd out.

```
// a pointer to a heap-allocated array of 10 integers, all 0
var x = alloc int64[10] : int64[10] *;

// a pointer to a heap-allocated struct with 0 fields.
var y = alloc struct { f : int64, g : bool }
      : struct { f : int64, g : bool } *;
```

When allocating pseudos or local variables of aggregate types on the stack, you will need to subtract a variable number of bytes from `RSP`. Thus, you need to change your compiler to track stack *offsets* rather than stack slot numbers for your pseudos and local variables. Moreover, since we require all such variables to be initialized with 0, you may want to make calls to `memset()` (or a similar function you write in your BX runtime) to zero out the stack. For example, see figure 1.

### 3.2 Referencing and Assignment

The key feature of an assignable is that it has a *memory location*. Variables are easy: they are just located at their corresponding data segment (for global variables) or stack slots (for local variables). For array subscripting (i.e., the `[]` notation), there is the additional cost of computing the offset into the array. Likewise, for structs it is necessary to compute the offset of the field that is accessed.

---

<sup>2</sup>Another planned feature of BX3 removed for time constraints.

```
f:
    pushq %rbp
    movq %rsp, %rbp
    subq $96, %rsp

    # now zero out the newly allocated stack frame
    pushq %rdi          # save old rdi (which may be arg1 of f)
    pushq %rsi          # save old rsi (which may be arg2 of f)
    pushq %rdx          # save old rdx (which may be arg3 of f)
    leaq 24(%rsp), %rdi  # top of stack ignoring above 3 saves
    movq 0, %rsi         # the thing to fill with, i.e., 0
    movq $96, %rdx       # the number of bytes to set
    callq memset         # the C standard library's memset()
    popq %rdx           # restore arg3 of f
    popq %rsi           # restore arg2 of f
    popq %rdi           # restore arg1 of f
    # done zeroing out the stack

    # ... rest of the function
```

Figure 1: Zeroing out the allocated stack frame

ERTL instruction	Description
Li: <b>load</b> $r_b, r_i, n_s, n_o \gg r_o \rightarrow L_o$	Load contents of memory location $r_b + r_i * n_s + n_o$ into $r_o$
Li: <b>store</b> $r, r_b, r_i, n_s, n_o \rightarrow L_o$	Store $r$ into memory location $r_b + r_i * n_s + n_o$ into

Figure 2: New (E)RTL instructions for loading/storing aggregates

One way to ease the compilation of memory accesses is to extend the (E)RTL instructions **load** and **store** as shown in figure 2. The computation  $r_b + r_i * n_s + n_o$  is actually rather natural for AMD64. The full syntax for memory addresses in instructions is: `offset(%rbase,%rindex,scale)` where `offset` is a literal 32-bit integer (or a label for PC-relative addressing), `scale`  $\in \{1, 2, 4, 8\}$ , and `%rbase` and `%rindex` are registers. All but `%rbase` are optional. The *effective address* of such an address is the value of the expression:

$$\%rbase + (\%rindex \times scale) + offset$$

Thus, for example, if **RAX** holds the address of an `int64*`  $e$ , and **RCX** holds the number  $n$ , then `0(%rax,%rcx,8)` is the address of  $e[n]$ .

For computing with addresses, an important instruction is **leaq** which computes an effective address and stores the address into the destination register. For instance:

```
leaq 0(%rax,%rcx,8), %r10    # R10 = RAX + RCX * 8 + 0
```

would set **R10** to the above computed value. Then, if you want to actually dereference the contents at that address, you might use **movq** like so: `movq (%r10), %r11`.

```

type fibtree = struct {
    val : int64,
    prev : fibtree*[2],
};

fun make_fibtree(n : int64) : fibtree* {
    var ret = alloc fibtree[1] : fibtree*;
    if (n == 0) {
        ret->val = 0;    // actually redundant
    } else if (n == 1) {
        ret->val = 1;
    } else {
        ret->prev[0] = make_fibtree(n - 1);
        ret->prev[1] = make_fibtree(n - 2);
        ret->val = ret->prev[0]->val + ret->prev[1]->val;
    }
    return ret;
}

proc main() {
    var ft : make_fibtree(20) : fibtree*;
    print ft->val;
}

```

Figure 3: Fibonacci Example in BX3